# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=haskell-src-exts-util
pkgname=haskell-src-exts-util
pkgver=0.2.5
pkgrel=274
pkgdesc="Helper functions for working with haskell-src-exts trees"
url="https://github.com/pepeiborra/haskell-src-exts-util"
license=("BSD")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-data-default' 'haskell-src-exts' 'haskell-uniplate')
makedepends=('ghc')
source=("https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz")
sha512sums=('0497d8e2206402cc295289b1893f326c8f5fa7fc00177279b7af389b3351a94a123d7eb754d8a245981bae51d1a978684791d9b111befca2888bc7fe22f07147')

prepare() {
    cd $_hkgname-$pkgver
    sed -i '/semigroups/d' $_hkgname.cabal
}

build() {
    cd $_hkgname-$pkgver

    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

package() {
    cd $_hkgname-$pkgver

    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
